



Dato- og tidsberegninger i Excel

En introduktion til simple beregninger med dato og tid i Excel.

Opdateret med afsnit om beregning af tillæg.

Skrevet den **02. feb 2009** af **jkrons** i kategorien **Regneark / Excel** | ★★★★★

Der kommer hen ad vejen temmelig mange spørgsmål der behandler beregninger med dato og tid. Derfor denne artikel, der prøver at beskrive nogle af de ting, man skal være opmærksom på ved dato/tidsberegninger i Excel.

Enheden:

Som udgangspunkt for alle Excels tidsberegninger er en enhed. Denne enhed er 1 døgn. Alle andre tidsangivelser måles ud fra denne enhed. Uger, måneder og år er således multipla af et døgn, mens timer, minutter og sekunder er brøkdele af et døgn.

1 år = 365 eller 366 døgn afhængigt af, om der er skudår.

1 måned = 28, 29, 30 eller 31 døgn, afhængigt af, hvilken måned, der er tale om.

1 uge = 7 døgn.

1 time = 1/24 døgn = 0,041666666666... døgn

1 minut = 1/1440 døgn = 0,0006944444... døgn

1 sekund = 1/86400 døgn = 0,000011547407407... døgn

Ud fra disse værdier udarbejdes alle dato- og tidsmæssige beregninger. Det er dog vigtigt at datoer og klokkeslæt er indtastet i et format, som Excel genkender som en dato eller et klokkeslæt. Datoer indtastes således som dd-mm-åå eller dd/mm/åå. Ved indtastning er det muligt at udelade foranstillede 0'er i dag og måned. Klokkeslæt indtastes som tt:mm:ss, altså med kolon som skilletegn mellem enhederne. Derimod kan man ikke bruge punktum, hverken i datoer eller klokkeslæt.

Da alle dato- og tidsværdier altså repræsenterer tal, kan man også bearbejde dem med almindelige regnefunktioner så som addition, subtraktion, multiplikation og division. Det er dog ikke alle operationer, der er lige nemme at foretage, men det hænger sammen med dato/tids formatet, ikke værdierne.

Datoberegninger:

En af de ofte forekommende opgaver er at trække to datoer fra hinanden. Dette gøres ret nemt. Har man fx startdatoen i A1, fx 12-02-05 og slutdatoen, fx 15-02-05 i B1, beregnes antal dage som =B1-A1. Resultatet vil så blive vist som 03-01-1900. Cellen skal nu formateres som tal, hvorefter resultatet 3 vises.

Årsagen til visningen er, at når to datoformaterede celler indgår i en formel, bliver resultatet også datoformateret. Når det lige bliver 03-01-1900 skyldes det, at Excel som standard anvender 1900 datosystemet, som tager udgangspunkt i 1. januar 1900. Dette er dag 1. Døgn nummer 3 er således 3. januar 1900, og 3 er antallet af dage mellem de to datoer. Læg mærke til, at den sidste dag også tælles med. Reelt set er der jo kun 2 dage MELLEMLIG de to datoer, og er det dette tal man er ude efter, kan man ændre sin formel til =B1-A1-1.

Hvis startdatoen er større end slutdatoen vises ikke noget resultat men #####. Dette skyldes at resultatet bliver negativt, og Excel kan ikke vise negative dato- eller tidsangivelser. Men hvis cellen formateres som tal, står der fint -3 med datoerne i ovenstående eksempel.

Man kan også lægge datoer sammen, gange dem med hinanden eller dividere dem med hinanden, men det har sjældent noget fornuftigt formål.

12-02-05 + 15-02-05 = 01-04-2110 eller 76.793

12-02-05 * 15-02-05 = ##### eller 1.474.291.210.

"Havelågerne" skyldes at Excel ikke kender så store datoer. Den største dato Excel kender er 31-12-9999, og det svarer til 2.958.465. Den mindste dato Excel kender er altså 01-01-1900, der svarer til 1. Indtaster man datoer før 1. januar 1900, vil de ikke blive opfattet som datoer, mens som tekst, og man kan derfor ikke bruge dem i beregninger.

12-02-05 / 15-02-05 = 00-01-1900

hvilket umiddelbart er meningsløst, da der jo ikke er en dag 00 i januar. Fjerner man formateringen vil man kunne se, at det svarer til 0,999921871, altså knap et døgn, og formaterer man dette tal som klokkeslæt svarer det så til 23:59:53.

Vi kan også kombinere beregninger med datoer og tal. Vi kan fx lægge dage til eller trække dem fra datoer.

12-02-05 + 2 = 14-02-05

12-02-05 - 2 = 10-02-05 osv.

Vi kan også gange og dividere, men igen er det ret meningsløst.

Excel har også en række indbyggede datofunktioner, som kan bruges til specielle udregninger. Dem skal jeg ikke komme ind på i denne forbindelse, men de kan fx bruges, hvis man skal lægge et antal måneder eller år til en dato. 12-2-05 + 2 måneder, eller 12-2-05 + 2 år kan fx løses med:

=DATO(ÅR(A1);MÅNED(A1)+A2;DAG(A1))

hvor A1 indeholder datoen, og A2 det antal måneder, der skal lægges til, eller

=DATO(ÅR(A1)+A2;MÅNED(A1);DAG(A1))

hvor A1 stadig er datoen, mens A2 er det antal år, der skal lægges til.

Klokkeslæts- eller tidsmæssige beregninger

Beregninger på klokkeslæt (tid) udføres på samme måde som datoberegninger. Forskellen er oftest, at i klokkeslætsberegninger vil vi gerne bevare klokkeslætsformatet.

Antag at

A1: 08:20:06

B1: 09:30:47

=B1-A1 giver 01:10:41

=B1+A1 giver 17:50:53

Vær opmærksom på, at hvis resultatet af en sammenlægning giver mere end 24, vises dette ikke som standard.

09:20 + 16:32 giver 01:52:00. For at få det vist som 25:52:00 skal cellen formateres med det brugerdefinerede format [t]:mm:ss.

Arbejdstidsregistrering

Arbejdstidsregistrering er ofte et ofte forekommende emne på Eksperten. Her skal man igen huske at Excel ikke kan vise negative klokkeslæt.

Mødetid Står i A1, Gå hjem tid i B1

Så længe mødetid altid ligger før gå hjem tid er der ingen problemer. Så kan arbejdstiden beregnes som $=B1-A1$. Men hvis man arbejder hen over midnat, altså at mødetiden ligger efter gå hjem tiden, vil denne formel ikke virke, da den jo så giver et negativt resultat. Problemet kan løses ved at ændre formlen til $=B1-A1+(A1>B1)$.

Forklaringen er at sidste parentes er et udsagn, der evalueres som sandt eller falsk. Prøv fx at skrive $=A1>B1$ i en celle. I det konkrete tilfælde evalueres udtrykket som sandt, fordi A1 er større end B1 (mødetid før gå hjem tid). Da SAND repræsenterer værdien 1, mens FALSK repræsenterer værdien 0, lægges der altså 1 til resultatet, hvis A1 er større end B1. 1 er grundenheden, altså et døgn.

Hvis vi bruger den oprindelige formel og formaterer som tal, kan vi se, at vi får et negativt tal. Har vi fx 23:37 som mødetid og 2:37 som gå hjem tid vil resultatet formateret som tal være -0,875. Lægger vi 1 til det, bliver resultatet 0,125. Dette svarer til 1/8 af et døgn eller 3 timer, netop den tid, der er arbejdet.

Skal man lave tidsberegninger, som strækker sig over mere end et kalenderdøgn, fx at man møder én dag kl. 22:00 og går hjem ikke næste dag, men næste dag igen kl. 02:00 er man nødt til at tage datoen med.

Så kan man fx indtaste mødetid som 01-02-05 22:00 og gå hjem tid som 03-02-05 02:00. Så kan man bruge formlen ovenfor, altså $=B1-A1+(A1>b1)$. Hvis man så formaterer cellen som [t]:mm:ss får man resultatet 28:00:00.

Lige som man kan lægge dage til en dato, kan man også lægge timer, minutter eller sekunder til et klokkeslæt. Har man fx 22:35 i A1 og vil lægge 1 time til, kan man bruge formlen: $=A1+1/24$. Skal man lægge 48 minutter til bliver formlen: $=A1+48/1440$ og skal man lægge 19 sekunder til skal formlen være $=A1+19/86400$.

Men hvad så, hvis man skal lægge 2 timer, 23 minutter og 46 sekunder til? Det kan gøre med $=A1+2/24+23/1440+46/86400$, men det nemmeste er at skrive $=A1+"2:23:46"$.

Beregninger at tillægsudløsende tid

Jeg er flere gange blevet spurgt om jeg ikke kunne forklare, hvordan man beregner overtid, nattillæg, aftentillæg osv. Dette har jeg veget tilbage for, da det er ret vanskeligt at sige noget konkret om, eftersom måden disse tillæg beregnes på, er meget forskellig fra virksomhed til virksomhed. Jeg har dog nu valgt at angive nogle forskellige eksempler, som man måske kan bygge videre på.

I alle eksemplerne går ud fra, at "mødetid" indtastes i A1, mens "gå-hjem tid" indtastes i B1.

Eksempel 1: Der udbetales tillæg, for timer, der arbejdes efter 17.

Opgaven går her alene ud på, at beregne, hvor mange timer, der arbejdes efter kl. 17.00. Og allerede her melder den første udfordring sig. Nemlig om der skal tages hensyn til mødetiden også - altså om man kan møde ind efter 17, og alligevel få tillæg for alle timer, eller om man skal have arbejdet et minimums antal timer, for at få tillæg for timerne efter 17. Desuden er der forskel på, om man kan arbejde efter midnat, eller om man altid stopper før midnat, samt om tillægget på et tidspunkt hører op igen. Så lige pludselig har eksemplet udviklet sig til indtil flere variationer:

Eksempel 1a: Uanset hvornår man møder, udbetales tillæg for alle timer efter 17. Dog kun til midnat. Det løses forholdsvis nemt med denne formel:

$=B1-MAKS(A1;(17/24))$

"Gå-hjem tid"- det tal, der er størst af henholdsvis mødetid eller kl. 17:00). Møder man kl. 16:00 og går hjem kl. 19:00 gives tillæg for 2 timer, fordi klokken 17 er senere end mødetidspunktet. Møder man kl. 18:00 og går hjem kl. 19:00 får man tillæg for 1 time, fordi mødetidspunktet er senere end kl. 17.

Eksempel 1b: Der udbetales tillæg for alle timer efter kl. 17. Også timer efter midnat. Dette er også forholdsvis simpelt. Ret formlen til:

$B1-MAKS(A1;(17/24))+(A1>B1)$

Der skal altså lægges 1 til, hvis "gå-hjem tid" ligger før mødetid, hvilket $(A1>B1)$ sørger for.

Eksempel 1c: Der udbetales tillæg efter kl. 17.00, men kun for timer, der overstiger dagens normaltimal (i eksemplet her sat til 7). Møder man således kl. 08:00 og går hjem kl. 18.00 udbetales 1 times tillæg. Møder man kl. 14.00 og går hjem kl. 18:00 udbetales ingen tillæg, fordi der ikke er arbejdet mere end 7 timer. Også denne situation kan have en variant, der går over midnat, men den må man selv arbejde sig frem til. Det burde være til at håndtere, hvis man ser på foregående eksempel.

$=HVIS(B1-A1>(7/24);HVIS(B1>17/24;B1-(17/24);0);0)$

løser problemet. Først undersøges, om der er arbejdet mere end 7 timer. I bekræftende fald beregnes tillægget for den tid, der ligger mellem kl. 17 og "gå hjem tiden"; i benægtende fald er tillægstiden 0:00.

Eksempel 2: Der udbetales tillæg for timer, der ligger før 06:00 og efter 17:00. Mødetid kl. 05:00 og "gå-hjem tid" kl. 19.00 skal således give 3 timer. Formlen er:

$=(B1<=A1)*(1-(17/24)+(6/24))+MIN((6/24);B1)-MIN((6/24);A1)+MAKS((17/24);B1)-MAKS((17/24);A1)$

Eksempel 3a: På lørdage, søndage og helligdage udbetales et ekstratillæg for timer mellem 11 og 17:

$=HVIS(A1>=(17/24);0;HVIS(A1>=(11/24);MIN(B1;(17/24))-A1;MIN(B1;(17/24))-(11/24)))$

Eksempel 3B: På lørdage, søndage og helligdage udbetales et forhøjet tillæg for timer efter kl. 17:

$=B1-MAKS(A1;(17/24))+(A1>B1)$

Og så videre. Mange af disse formler vil give fejl, hvis det ene eller begge felter ikke er udfyldt. De kan derfor alle suppleres med test for, om et eller begge felter er tomme, og der er indtastet tal mm. Eksempel 3a, med sådanne test indbyggede, kunne så se således ud:

$=HVIS(ER.TAL(HVIS(ELLER(ER.TOM(A1);ER.TOM(B1))=SAND;"");(B1<=A1)*(1-(17/24)+(6/24))+MIN((6/24);B1)-MIN((6/24);A1)+MAKS((17/24);B1)-MAKS((17/24);A1)));HVIS(ELLER(ER.TOM(A1);ER.TOM(B1))=SAND;"");(B1<=A1)*(1-(17/24)+(6/24))+MIN((6/24);B1)-MIN((6/24);A1)+MAKS((17/24);B1)-MAKS((17/24);A1)))$

De ovenstående eksempler tager udgangspunkt i mødetid/"gå-hjem tid", men man kan også forestille sig situationer, hvor det alene er varigheden af arbejdet, der udløser tillæg, uafhængigt af, hvornår man arbejder - eller evt. i kombination med arbejdstiden.

De faste tider, fx 6, 11 og 17, kan selvfølgelig skiftes ud med andre. Eller måske bedre, de kan erstattes med cellereferencer, hvor standardtiderne så kan indtastes.

Som skrevet ovenfor, er mulighederne legio, så den nemmeste måde at få løst et konkret problem på, er enten at prøve sig frem med Excels mange muligheder, eller at beskrive sit problem detaljeret, og så stille spørgsmålet under Regneark.

Som det kan ses er der mange muligheder for tidsberegninger, men det vigtigste er at huske grundtallene,

nævnt først i artiklen.

1904-datosystemet

Nu har jeg flere gange nævnt at Excel ikke kan vise negative datoer eller klokkeslæt, og det er også korrekt, hvis man anvender 1900-datosystemet, som er standard i Excel. Men skifter man til 1904-datosystemet, kan man godt vise negative datoer og klokkeslæt.

Men skifter ved at vælge Funktioner - Indstillinger. I fanebladet beregning sættes flueben i 1904-datosystem. Nu kan negative datoer og klokkeslæt vises. Vær dog opmærksom på, at allerede indtastede datoer nu ændrer sig. For eksempel vil 19-04-2005 blive til 20-04-2009.

Det skyldes at i stedet for at anvende 01-01-1900 som udgangsdato, anvendes nu 02-01-1904.

Tilføjelse: Excel kan faktisk godt vise hele negative timer minutter eller sekunder. Formateres cellen som [tt] vises hele timer, også negative. [mm] giver hele minutter, mens [ss] giver hele sekunder. Desværre kan man ikke kombinere fx [tt]:[mm] eller [tt:mm]. Det tillades ikke.

Kommentar af erlandsen d. 08. jul 2005 | 1

Udmærket artikel - så mangler vi bare lidt om beregning af nattillæg, overtidstillæg osv...

Kommentar af halifax d. 10. nov 2007 | 2

Selv om jeg aktuelt ikke har brug for dato- og tidsberegning, er det en fornøjelse at se andre stille deres viden til rådighed, endda med gode eksempler. Det vil jeg gerne påskønne.

Kommentar af mugs d. 05. maj 2005 | 3

Beregninger med datoer volder tit problemer i både Excel og Access. Denne artikel forklarer kort og godt hvordan det skal gøres i Excel. Udmærket artikel og initiativ. Der bør dog rettes et par slåfejl. / mugs

Kommentar af bak d. 22. apr 2005 | 4

Super artikel. Den skal nok blive flittig besøgt / henvist til :-)

Kommentar af b_hansen d. 30. okt 2006 | 5

I betragtning af, at jeg ofte henviser til denne gode artikel, er det vel på sin plads, at jeg også giver bedømmelse her. Det er en god artikel, der gør tidsberegninger meget lettere. /b_hansen

Kommentar af espersen d. 01. maj 2006 | 6

nej dog - flot samling af tips til excel-udregninger. Lækkert 'opslagsværk'

Kommentar af Iklingenberg d. 23. aug 2006 | 7

Nice!

Kommentar af splazz d. 23. aug 2007 | 8

den ryger direkte ind i foretrukne! :)

Kommentar af erikjuul d. 09. jul 2005 | 9

God forklaring af sammenhængen i dato og klokkeslets-beregninger.

Kommentar af cc.lassen d. 23. maj 2005 | 10

Kommentar af nielle d. 05. okt 2008 | 11

Super artikel :)

Kommentar af natkatten d. 28. jul 2008 | 12

Kommentar af miko67 d. 21. apr 2005 | 13

En rigtig - RIGTIG - god artikel. /miko67

Kommentar af jlemming d. 17. okt 2007 | 14

God artikle, med mange gode eksempler. skifter godt mellem pratisk og teori

Kommentar af p-j-dk d. 09. apr 2006 | 15

Lige hvad jeg søgte!

Kommentar af thonis d. 03. jan 2007 | 16

Kommentar af pazau d. 17. okt 2008 | 17

Artiklen viser hvordan man liver tids og datoberegninger i Excel med eksempler. En super artikel for alle og enhver. Den er nem at forstå, og eksemplerne gør det ekstra let at forstå artiklen. Alt i alt kunne den ikke være bedre. :)

Kommentar af bdc d. 20. mar 2010 | 18

det tog lidt tid for mig at forstå det, men så fandt jeg ud af det.
super guidexD

Kommentar af eskape d. 12. maj 2010 | 19

Rigtig god guide.

Kan jeg lægge "6 måneder" til en dato? Altså, i stedet for at sige dato+6x30, så kan excel selv lægge 28-30-31 dage til pr. måned, så man får præcist 6 måneder efter den nævnte dato?

Kommentar af eskape d. 12. maj 2010 | 20

Dumme mig - jeg skal jo bare lære at læse istedet for at skimme.

Jeg vil gerne gentage - god guide. Og hvordan sletter man sine dumme kommentarer? (hehe)

Kommentar af mike23 d. 29. jun 2010 | 21

Perfekt for en excel-newbie som mig :)

Kommentar af AndersWested d. 04. jul 2010 | 22

Det kan tage lang tid at lave et system til brugervenlig indtastning af tider i excel, altså den første del før man kommer i gang med beregningerne.

Derfor har jeg kreeret en hjemmesideløsning til dette: www.arbejdstider.dk, hvorfra man senere kan eksportere sine tidsata til regneark og bearbejde dem som man lyster.

Der er to faciliteter til eksport: Enten Filer -> Eksport eller indberet løntimer. Det sidste benyttes, når folk selv indtaster deres arbejdstimer og man senere samler data sammen fra alle ansatte i et enkelt regneark.

Kommentar af oyejo d. 01. okt 2010 | 23

Det er en gammel artikel, men meget god. Legger en kommentar slik at jeg kan finne tilbage ;-)

Kommentar af September d. 22. aug 2011 | 24

Når man så har, hvor mange timer der er overarbejde, hvordan * dette så med beløbet? altså tid * tal(beløb). for mig giver det nogle mærkelige tal

Kommentar af mrsoerennielsen d. 27. feb 2012 | 25

Super

Kommentar af finb d. 29. sep 2012 | 26

Tak

Kommentar af Zem1985 d. 03. apr 2013 | 27

Dette er et super indlæg, men jeg har en yderligere udfordring.

Jeg har lavet et tidsskema for nogle af vores medarbejdere, som automatisk skal have fratrukket 30 min til frokost, hver gang de har en arbejdsdag mere end 5 timer. Er der nogen som kan hjælpe med det?

På forhånd tak.

Kommentar af finb d. 19. jun 2013 | 28

til zem:

=hvis(timer>5;timer-(1/48);timer)

Kommentar af OleMo d. 24. aug 2014 | 29

Super indlæg, men jeg finder desværre ikke svar på følgende:

Til en vagtplan i en kalender i Excel ville jeg gerne kunne markere hver 3. weekend, dvs hver 3. lørdag og søndag.

Årets første vagt-weekend er fx lørdag og søndag i uge 2.

Dvs næste vagt-weekend bliver uge 5, uge 8 o.s.v.

Jeg bruger denne til beregning af ugenumrene:

=AFKORT((B1-DATO(ÅR(B1+3-REST(B1-2;7));1;REST(B1-2;7)-9))/7)

Jeg ville fx gerne kunne lave en kolonne, hvor der ud for hver lørdag og søndag kommer til at stå hhv Fri eller Arbejde.

Håber nogen kan hjælpe mig.

På forhånd tak.

Kommentar af souldr d. 17. dec 2015 | 30

Tak for en god artikel, og uden at være ekspert udi excel kan man find mange god oplysninger. Jeg svaner dog en metode til at beregne overtid når man kun "må" registrerer overtidstimer, altså de timer man arbejder over det normal arbejdstid og oven i købet til 2 forskellige satser.

Kommentar af EigilKousgaard d. 28. feb 2016 | 31

Jeg har vist nok et meget simpelt spørgsmål, men jeg kan ikke selv få det til at fungere, så jeg hænger nok lidt i bremsen :-). Hvordan lægger man tid (timer og minutter) til et klokkeslæt....?

hilsen Kousgaard

Kommentar af EigilKousgaard d. 28. feb 2016 | 32

jeg skal lige pointere at det er manuelt indtastet tid jeg ikke kan lægge til et klokkeslæt, hvis jeg indtaster tiden i en celle og lægger de to celler sammen så virker det fint, men nu har jeg brug for at tillægge tid manuelt....:-|